
20060396_readme.pdf


Readme
Data and do files for “Indirect Effect of an Aid Program: How do Cash 


Transfers Affect Ineligibles’ Consumption?”, 
by Manuela Angelucci and Giacomo De Giorgi


Tables 3, 4: and Figure 2:


Do-file: AER_credit.do
Data: AER_credit.dta


Table 8:


Do-file: grains_submitted.do
Data: grains_submitted.dta


All other Tables:
Do file: table.do
Each data set is named according to the corresponding Table (e.g. the data for Table 5 are 
called table5.dta).
The two table6.dta files (separate data sets for the 1998 and 1999 data) are also needed 
for Table 9.


To merge different data set in Stata, use the following syntax:


merge folio time using “using.dta”


Folio is the unique household identifier is folio, while t is the time variable (t=7,8,9,10 
for Sept. 97, Nov. 98, May 99, and Nov. 99).







AER_credit.do
clear
cap log close
set more off
set mem 500m
set matsize 800
version 8.2
use AER_credit.dta

/* TABLE 3*/



**************************************
**** FOR PROPORTIONS WE USE THE WHOLE SAMPLE
**** WHILE FOR QUANTITIES WE TRIMMED THE LAST PERCENTILE
*****************************************************


foreach bbb in treatnp treatp {

preserve
keep if `bbb'!=. & t==8


tab `bbb' if tot_creditq_r_ae<=10000, sum(tot_creditq_r_ae)

tab `bbb' if tot_creditq_r_ae>0 & tot_creditq_r_ae<=10000, sum(tot_creditq_r_ae)

g pos=1 if tot_creditq_r_ae>0 & tot_creditq_r_ae!=.
replace pos=0 if pos==. &  tot_creditq_r_ae!=.

tab `bbb', sum(pos)

restore
}


*****************************************************************************
****************          FORMAL LOANS                      *****************
*****************************************************************************


foreach bbb in treatnp treatp {
forvalues i=8/9{


preserve
keep if `bbb'!=. & t==`i'

noi di `bbb' "time"  `i'

tab `bbb'  if loanq_r_ae<20000 , sum(loanq_r_ae)

tab `bbb' if loanq_r_ae>0  & loanq_r_ae<20000, sum(loanq_r_ae)

g pos=1 if loanq_r_ae>0 & loanq_r_ae!=.
replace pos=0 if pos==. & loanq_r_ae!=.

tab `bbb', sum(pos)

restore
}
}

*****************************************************************************
****************         FF TRANSFERS                     *****************
*****************************************************************************

foreach var in fftransferq_r_ae{
foreach bbb in treatnp treatp {
forvalues i=8(2)10{

preserve
keep if  `bbb'!=. & t==`i'

noi di `bbb' "time"  `i'

tab `bbb' if `var'<2500, sum(`var')
tab `bbb' if `var'>0 & `var'<2500, sum(`var')

g pos=1 if `var'>0 & `var'!=.
replace pos=0 if pos==. & `var'!=.

tab `bbb', sum(pos)


restore
}
}
}




*****************************************
********** TRANSFERS IN KIND***********
*****************************************

foreach var of varlist fftransferdkind  {
foreach bbb of varlist treatnp treatp {
forvalues i=8(2)10{

preserve
keep if `var'!=. & `bbb'!=. & t==`i'

noi di `var'  `bbb' "time"  `i'

tab `bbb' , sum(`var')

restore
}
}
}

*****************************************
********** TABLE 4  NO CONTROLS***********
*****************************************


foreach var in np p{
foreach bbb in loanq_r_ae{
forvalues i=8(1)9{

char treat`var' [omit] 0

di "YEAR" `i'

preserve

keep if t==`i' & treat`var'!=. & `bbb'!=.

xi: dprobit loand i.treat`var'  if t==`i', r cl(village)

xi: reg `bbb' i.treat`var' if t==`i' & `bbb'<=20000, r cl(village)

xi: tobit `bbb' i.treat`var'  if t==`i' & `bbb'<=20000, ll(0)
dtobit


restore

}
}
}


foreach var in np p{
forvalues i=8(2)10{

char treat`var' [omit] 0

di "YEAR" `i'

preserve

keep if t==`i' & treat`var'!=.

xi: dprobit fftransferq_r_ae i.treat`var'   if t==`i', r cl(village)

xi: reg fftransferq_r_ae i.treat`var' if t==`i' & fftransferq_r_ae<=2500, r cl(village)

xi: tobit fftransferq_r_ae i.treat`var' if t==`i' & fftransferq_r_ae<=2500, ll(0)
dtobit

restore

}
}



foreach var in np p{
foreach bbb in fftransferdkind{
forvalues i=8(2)10{

char treat`var' [omit] 0

di "YEAR" `i'

preserve

keep if t==`i' & treat`var'!=.

xi: dprobit `bbb' i.treat`var' if t==`i', r cl(village)

restore

}
}
}



/* APPENDIX: TABLE 4 ADDING CONTROLS */


recode hhhsex 9=.
foreach var in np p{
foreach bbb in loanq_r_ae{
forvalues i=8(1)9{

char treat`var' [omit] 0

di "YEAR" `i'

preserve

keep if t==`i' & treat`var'!=. & `bbb'!=.

xi: dprobit loand i.treat`var' avg_sh indice vhhnum yycali i.hhhsex  i.p16 i.hhhalpha hhhage hectareas if t==`i', r cl(village)

xi: reg `bbb' i.treat`var' avg_sh indice vhhnum yycali i.hhhsex  i.p16 i.hhhalpha hhhage hectareas  if t==`i' & `bbb'<=20000, r cl(village)

xi: tobit `bbb' i.treat`var'  avg_sh indice vhhnum yycali i.hhhsex  i.p16 i.hhhalpha hhhage hectareas if t==`i' & `bbb'<=20000, ll(0)
dtobit


restore

}
}
}


foreach var in np p{
forvalues i=8(2)10{

char treat`var' [omit] 0

di "YEAR" `i'

preserve

keep if t==`i' & treat`var'!=.

xi: dprobit fftransferq_r_ae i.treat`var'  avg_sh indice vhhnum yycali i.hhhsex  i.p16 i.hhhalpha hhhage hectareas if t==`i', r cl(village)

xi: reg fftransferq_r_ae i.treat`var' avg_sh indice vhhnum yycali i.hhhsex  i.p16 i.hhhalpha hhhage hectareas if t==`i' & fftransferq_r_ae<=2500, r cl(village)

xi: tobit fftransferq_r_ae i.treat`var' avg_sh indice vhhnum yycali i.hhhsex  i.p16 i.hhhalpha hhhage hectareas  if t==`i' & fftransferq_r_ae<=2500, ll(0)
dtobit

restore

}
}



foreach var in np p{
foreach bbb in fftransferdkind{
forvalues i=8(2)10{

char treat`var' [omit] 0

di "YEAR" `i'

preserve

keep if t==`i' & treat`var'!=.

xi: dprobit `bbb' i.treat`var' avg_sh indice vhhnum yycali i.hhhsex  i.p16 i.hhhalpha hhhage hectareas  if t==`i', r cl(village)

restore

}
}
}

/* FIGURE 2*/


/*ADULT EQUIVALENT CONSUMPTION BY YYCALI AND WAVE*/

graph twoway lowess Cind yycali if t==9 & treatnp==1 & inrange(yycali,600,1200), bw(30) lc(black) lp(solid) legend(label(1 "Treated")) ytick(none)  /*
*/ ||  lowess Cind yycali if t==9 & treatnp==0 & inrange(yycali,600,1200), bw(30) lc(black) lp(dot)  legend(label(2 "Control")) ytick(none)  xtitle(Welfare) ytitle(Consumption) title(1999 May)

graph save Cind_yycali_9.gph, replace

graph twoway lowess Cind yycali if t==10 & treatnp==1 & inrange(yycali,600,1200), bw(30) lc(black) lp(solid) legend(label(1 "Treated")) /*
*/ ||  lowess Cind yycali if t==10 & treatnp==0 & inrange(yycali,600,1200), bw(30) lc(black) lp(dot) legend(label(2 "Control")) xtitle(Welfare) ytitle(Consumption) title(1999 November)

graph save Cind_yycali_10.gph, replace

graph combine Cind_yycali_9.gph Cind_yycali_10.gph, row(1) iscale(1) saving(Cind_yycali.gph, replace)

forvalues t=9/10{
erase Cind_yycali_`t'.gph
}

log close




AER_credit.dta


grains.dta


grains_submitted.do
****************************************************
******THIS DO FILE PRODUCES TABLE 9 IN THE   *******
****** PAPER. DO FILE BY MANUELA ANGELUCCI   *******
******  AND GIACOMO DE GIORGI                *******
******                                       *******
****************************************************


clear
set more off
capture log close
set mem 800m

use grains.dta, clear

/* PRODUCTION MINUS SALES*/
/*CORN, MEASURED IN MONTHLY PER ADULT EQUIVALENT KILOGRAMS*/

bys time: reg item1_p_s_ae NP, cl(village)
bys time: reg item1_p_s_ae P , cl(village)

/*BEANS, MEASURED IN MONTHLY PER ADULT EQUIVALENT KILOGRAMS*/

bys time: reg item2_p_s_ae NP, cl(village)
bys time: reg item2_p_s_ae P , cl(village)



/* CONSUMPTION*/
/*CORN, MEASURED IN MONTHLY PER ADULT EQUIVALENT KILOGRAMS*/

bys time: reg item1_hp_ae NP, cl(village)
bys time: reg item1_hp_ae P , cl(village)

/*BEANS, MEASURED IN MONTHLY PER ADULT EQUIVALENT KILOGRAMS*/

bys time: reg item2_hp_ae NP, cl(village)
bys time: reg item2_hp_ae P , cl(village)


/* CONSUMPTION*/
/*CORN, MEASURED IN MONTHLY PER ADULT EQUIVALENT KILOGRAMS*/

bys time: reg value_item1_hp_ae NP, cl(village)
bys time: reg value_item1_hp_ae P , cl(village)

/*BEANS, MEASURED IN MONTHLY PER ADULT EQUIVALENT KILOGRAMS*/

bys time: reg value_item2_hp_ae NP, cl(village)
bys time: reg value_item2_hp_ae P , cl(village)




table1.dta


table2.dta


table5.dta


table6_98.dta


table6_99.dta


table7.dta


tables.do
clear
cap log close
set more off
set matsize 800
set mem 100m





cd "c:\Consumption\Revision2007"


log using c:\Consumption\Revision2007\ResubmissionFeb2008\table.log, replace


********************************************************
***************         Table 1           **************
********************************************************


u "ResubmissionFeb2008\table1.dta", clear


/* ANALYSIS WITHOUT CONTROLS*/


foreach bbb in C { 
qui xi: reg `bbb'ind
outreg using "fcons`bbb'" , nolabel se bdec(2)  bracket 3aster nonobs nor2 replace

foreach var in np p { 
table treat`var' t if Cind<10000, c(m `bbb'ind sd `bbb'ind) f(%9.2f)

forvalues i=8/10{
di "YEAR" `i'



char treat`var' [omit] 0
xi: reg `bbb'ind i.treat`var' if t==`i' & Cind<10000, r cl(village)         

}
}
}

foreach var in np p { 
forvalues i=8/10{
xi: reg Cind treat`var' avg_s hhhage i.hhhsex  i.p16 i.hhhalpha i.hhhspouse yycali_1 i.region_1 indice hectareas vhhnum if t==`i' & Cind<10000, cl(village)
}
}



********************************************************
***************         Table 2           **************
********************************************************
u "ResubmissionFeb2008\table2.dta", clear


foreach p in n p {
xi: reg tomato treat`p' avg_s hhhage i.hhhsex  i.p16 i.hhhalpha i.hhhspouse yycali_1 i.region_1 indice hectareas vhhnum, cl(village)
}

foreach v in zana hoja nara manz limo pollo carne meat pesc huevo leche queso tort maiz trigo arro frij {
foreach p in n p {
xi: reg `v' treat`p' avg_s hhhage i.hhhsex  i.p16 i.hhhalpha i.hhhspouse yycali_1 i.region_1 indice hectareas vhhnum, cl(village)
}
}


********************************************************
***************         Table 5           **************
********************************************************

u "ResubmissionFeb2008\table5.dta", clear

foreach var in np p {
xi: reg newhhlab  i.treat`var'*i.t, r cl(village)         
}

********************************************************
***************      Tables 6 and 9       **************
********************************************************

u "ResubmissionFeb2008\table6_98.dta", clear

foreach bbb in np p {
preserve
keep if treat`bbb'!=.

    foreach depvar of varlist cosecha_kept cosecha_ton cosecha_soldton animals animalp animalnets r163 r165 producenetsale  {

    replace `depvar'=`depvar'/hhwsize7398  /* adult equivalent */
    _pctile `depvar', percentile(1,99)
    replace `depvar'=. if (`depvar'<r(r1) | `depvar'>r(r2))
    bys treat`bbb': su `depvar', d

    xi: reg `depvar' treat`bbb', r cl(village)
    }
    foreach depvar in cosecha cosecha_sold animals animalp agrcost grosssale {


    su `depvar'd if treat`bbb'==0

    xi: dprobit `depvar'd treat`bbb', r cl(village)
    }
    
restore
}

 


u "ResubmissionFeb2008\table6_99.dta", clear

foreach bbb in np p {
preserve
keep if treat`bbb'!=.

    foreach depvar of varlist cosecha_kept cosecha_ton cosecha_soldton r163 r165 producenetsale animals animalp animalnets {

    replace `depvar'=`depvar'/hhwsize7399m  /* adult equivalent */
    _pctile `depvar', percentile(1,99)
    replace `depvar'=. if (`depvar'<r(r1) | `depvar'>r(r2))
    bys treat`bbb': su `depvar', d

    xi: reg `depvar' treat`bbb', r cl(village)
    
    }

    foreach depvar in cosecha cosecha_sold animals animalp grosssale agrcost {

    su `depvar'd if treat`bbb'==0

    xi: dprobit `depvar'd treat`bbb', r cl(village)
    
    }
restore
}






********************************************************
***************          Table7           **************
********************************************************


u "ResubmissionFeb2008\table7.dta", clear
egen id=group(enti mpio local folio)
tsset id t

foreach depvar in horse donkey ox goat cow chicken pig rabbit {
g d`depvar'=d.`depvar'
replace d`depvar'=d`depvar'/14 if t==8
replace d`depvar'=d`depvar'/6 if t==9
replace d`depvar'=d`depvar'/6 if t==10

foreach bbb in treatn treatp {

                forvalues t=8/10 {
                reg d`depvar' `bbb' if t==`t', r cl(village)
                }
        }
}





log close



